Уважаемые гуру, ткните носом, что делаю не так?

Платформа 8.3.15.1830. Файловая ИБ.
Пытаюсь сделать простенький отчет с получением данных из внешней ИБ.
Подключаюсь - нормально, делаю запрос - тоже нормально, формирую табличный документ - все хорошо. Дальше передаю табличный документ на сторону клиента, чтобы собственно вывести отчет и получаю:
 Ошибка отображения типов:
Отсутствует отображение для типа 'COMОбъект'..

Вот код, вроде никакого криминала:

```1c
    &НаСервере
    Процедура СформироватьНаСервере( ТабДок)

    //"File="D:\BUH30"; Usr="administrator";Pwd="123456"";
    СтрокаПодключения = "File=""+ПутьКБазе+"";" + " Usr=""+ Логин+"";"+ "Pwd=""+ Пароль+"";
    Подключение = Новый COMОбъект("V83.COMConnector");

    Попытка
    База = Подключение.Connect(СтрокаПодключения);
    Исключение
    Сообщить(ОписаниеОшибки());
    База = Неопределено;
    Возврат;
    КонецПопытки;

    Запрос = База.NewObject("Запрос");

    Запрос.Текст =
    "ВЫБРАТЬ
    | ВзаиморасчетыСКонтрагентамиОстатки.Контрагент КАК Контрагент,
    | ВзаиморасчетыСКонтрагентамиОстатки.Контрагент.Наименование КАК КонтрагентНаименование,
    | СУММА(ВзаиморасчетыСКонтрагентамиОстатки.СуммаОстаток) КАК Долг
    |ИЗ
    | РегистрНакопления.ВзаиморасчетыСКонтрагентами.Остатки КАК ВзаиморасчетыСКонтрагентамиОстатки
    |
    |СГРУППИРОВАТЬ ПО
    | ВзаиморасчетыСКонтрагентамиОстатки.Контрагент
    |ИТОГИ
    | СУММА(Долг)
    |ПО
    | ОБЩИЕ";

    РезультатЗапроса = Запрос.Выполнить();

    Макет = РеквизитФормыВЗначение("Объект").ПолучитьМакет("Макет");
    ОбластьЗаголовок = Макет.ПолучитьОбласть("Заголовок");
    ОбластьПодвал = Макет.ПолучитьОбласть("Подвал");
    ОбластьШапкаТаблицы = Макет.ПолучитьОбласть("ШапкаТаблицы");
    ОбластьПодвалТаблицы = Макет.ПолучитьОбласть("ПодвалТаблицы");
    ОбластьОбщийИтог = Макет.ПолучитьОбласть("ОбщиеИтоги");
    ОбластьДетальныхЗаписей = Макет.ПолучитьОбласть("Контрагент");

    ТабДок.Очистить();
    ТабДок.Вывести(ОбластьЗаголовок);
    ТабДок.Вывести(ОбластьШапкаТаблицы);
    ТабДок.НачатьАвтогруппировкуСтрок();

    ВыборкаОбщийИтог = РезультатЗапроса.Выбрать(База.ОбходРезультатаЗапроса.ПоГруппировкам);

    ВыборкаОбщийИтог.Следующий(); // Общий итог
    ОбластьОбщийИтог.Параметры.Заполнить(ВыборкаОбщийИтог);
    ТабДок.Вывести(ОбластьОбщийИтог, ВыборкаОбщийИтог.Уровень());

    ВыборкаДетальныеЗаписи = ВыборкаОбщийИтог.Выбрать(База.ОбходРезультатаЗапроса.ПоГруппировкам);

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
    ОбластьДетальныхЗаписей.Параметры.Заполнить(ВыборкаДетальныеЗаписи);
    ТабДок.Вывести(ОбластьДетальныхЗаписей, ВыборкаДетальныеЗаписи.Уровень());
    КонецЦикла;

    ТабДок.ЗакончитьАвтогруппировкуСтрок();
    ТабДок.Вывести(ОбластьПодвалТаблицы);
    ТабДок.Вывести(ОбластьПодвал);

    База = Неопределено;

    КонецПроцедуры

    &НаКлиенте
    Процедура Сформировать(Команда)
    ТабДок = Новый ТабличныйДокумент;
    СформироватьНаСервере(ТабДок);
    ТабДок.Показать();
    КонецПроцедуры
```

Где оно умудряется COMОбъект на клиент протащить не понимаю..
---
Где оно умудряется COMОбъект на клиент протащить
Вероятно, здесь
```1c
    ОбластьДетальныхЗаписей.Параметры.Заполнить(ВыборкаДетальныеЗаписи);
```
 через параметр "Контрагент".
---
Где оно умудряется COMОбъект на клиент протащить
Вероятно, здесь
```1c
    ОбластьДетальныхЗаписей.Параметры.Заполнить(ВыборкаДетальныеЗаписи);
```
 через параметр "Контрагент".
---
Спасибо, убрал из запроса поле Контрагент и все взлетело